Wind Tunnel

Let us note k = K [a, k, ω, ζ] and let us define the Ore algebra A = K(t)[∂, δ], where ∂ is ordinary differential operator defined by $\partial f(t) = f'(t)$ and δ is a shift operator defined by $\delta f(t) = f(t+1)$.

```
A = OreAlgebra[Der[t], S[-1][t], a, k, ω, ζ]
```

```
𝕂(t)[a, k, ω, ζ][Dt; 1, Dt][S[-1][t]; #1 /. t → t - 1 &, 0 &]
```

We consider the system defined by the matrix $R \in A^{3\times4}$ given by

```
mat = {{Der[t] + a, -k a S[-1][t], 0, 0},
   {0, Der[t], -1, 0}, {0, ω^2, Der[t] + 2 ζ ω, -ω^2}};
MatrixForm[R = ToOrePolynomial[mat, A]]
```

$$\begin{pmatrix} D_t + a & -S[-1][t]\, a\, k & 0 & 0 \\ 0 & D_t & -1 & 0 \\ 0 & \omega^2 & 2\,\omega\,\zeta + D_t & -\omega^2 \end{pmatrix}$$

and the left A-module, finitely presented by R, is M = $A^{1\times4} / A^{1\times3} R$.
Let us compute the obstruction for M to be projective

```
P = ObstructionToProjectiveness[R, A]
```

$$\{\{\{-D_t\,\omega^2 - a\,\omega^2\}, \{D_t^3 + 2\,D_t^2\,\omega\,\zeta + D_t^2\,a + 2\,D_t\,a\,\omega\,\zeta\}, \{D_t^3\,\omega + D_t^2\,a\,\omega\}, \{D_t^4 + D_t^3\,a\}, \{(S_t^{-1})\,a\,k\,\omega^2\}\}\}$$

Let us deduce the obstructions in δ for M to be a projective left A-module.

```
obs = OreIntersection[Flatten[P], {S[-1][t]}, A]
```

$$\{(S_t^{-1})\}$$

Hence we obtain that B ⊗ M is projective over the ring B = A⟨σ⟩ / ⟨$\delta\,\sigma$ -1, $\sigma\delta$ - 1⟩, where σ is defined by σf(t)=f(t+1) (i.e. the two sided inverse of δ).

```
AnT = AnnihilatorTorsionModule[R, A]
```

```
Module is not torsion
```

Let us compute the paramterization of the system

```
MatrixForm[param = Parametrization[R, ξ, A]]
```

$$\begin{pmatrix} -a\,k\,\omega^2\,\xi[1][-1+t] \\ -\omega^2\,(a\,\xi[1][t] + \xi[1]'[t]) \\ -\omega^2\,(a\,\xi[1]'[t] + \xi[1]''[t]) \\ -a\,\omega^2\,\xi[1][t] - \omega\,(2\,a\,\zeta + \omega)\,\xi[1]'[t] - a\,\xi[1]''[t] - 2\,\zeta\,\omega\,\xi[1]''[t] - \xi[1]^{(3)}[t] \end{pmatrix}$$

We note that this parametrization is a minimal one:

```
MatrixForm /@ (L = MinimalParametrizations[R, A])
```

$$\left\{\begin{pmatrix} -(S_t^{-1})\,a\,k\,\omega^2 \\ -D_t\,\omega^2 - a\,\omega^2 \\ -D_t^2\,\omega^2 - D_t\,a\,\omega^2 \\ -2\,D_t^2\,\omega\,\zeta - 2\,D_t\,a\,\omega\,\zeta - D_t^3 - D_t^2\,a - D_t\,\omega^2 - a\,\omega^2 \end{pmatrix}\right\}$$

Let us check whether or not L admits a left inverse:

```
LeftInverse[L[[1]], A]
```

```
{}
```

We obtain that L is not an injective parametrization over A. Let now define the following Ore algebra

```
B = OreAlgebra[Der[t]]
```

$\mathbb{K}(t)[D_t; 1, D_t]$

and if we substitute S[-1][t] by δ, i.e.

```
L1 = L[[1]] /. S[-1][t] → δ
```

$$\{\{-\delta\, a\, k\, \omega^2\}, \{-D_t\, \omega^2 - a\, \omega^2\}, \{-D_t^2\, \omega^2 - D_t\, a\, \omega^2\}, \{-2\, D_t^2\, \omega\, \zeta - 2\, D_t\, a\, \omega\, \zeta - D_t^3 - D_t^2\, a - D_t\, \omega^2 - a\, \omega^2\}\}$$

```
L2 = ChangeOreAlgebra[L1, B]
```

$$\{\{-a\, k\, \delta\, \omega^2\}, \{-\omega^2\, D_t - a\, \omega^2\}, \{-\omega^2\, D_t^2 - a\, \omega^2\, D_t\}, \{-D_t^3 + (-a - 2\, \zeta\, \omega)\, D_t^2 + (-2\, a\, \zeta\, \omega - \omega^2)\, D_t - a\, \omega^2\}\}$$

then we obtain that the matrix L2 admits the following left inverse:

```
MatrixForm[T = Factor[iOrePolynomialToNormal[LeftInverse[L2, B]]]]
```

$$\begin{pmatrix} -\frac{1}{a\, k\, \delta\, \omega^2} & 0 & 0 & 0 \end{pmatrix}$$

Therefore if we use the advance operator $\sigma = \delta^{-1}$ then flat outputs of the system are defined by $\xi = T(x_1, x_2, x_3, u)$.

Let us try to decompose the system. We first compute some endomorphisms of M

```
P = Morphisms[R, R, {1, 0}, p, A]
```

Solve::svars : Equations may not give solutions for all "solve" variables. >>

$$\{\{p[3][1]\, D_t + p[2][1]\, (S_t^{-1}) + p[1][1], 0, 0, 0\}, \{0, p[2][1]\, (S_t^{-1}) + p[1][1], p[3][1], 0\}, \{0, 0, p[3][1]\, D_t + p[2][1]\, (S_t^{-1}) + p[1][1], 0\}, \{0, 0, 0, p[3][1]\, D_t + p[2][1]\, (S_t^{-1}) + p[1][1]\}\}$$

Let us try to see if among the above endomorphisms we can find some nontrivial idempotents

```
MatrixForm /@ (PAll = IdempotentMorphisms[P, R, p, A])
```

$$\left\{\begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}, \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}\right\}$$

We find that the only idempotents among the above endomorphisms are the trivial ones. Let us then try to find Λ such that ΛRΛ=-Λ so that P=I+ΛR defines a nontrivial idempotent:

```
P2 = PAll[[2]]
```

```
{{1, 0, 0, 0}, {0, 1, 0, 0}, {0, 0, 1, 0}, {0, 0, 0, 1}}
```

```
MatrixForm[ric = Riccati[R, P2, {0, 0}, p, A]]
```

Solve::svars : Equations may not give solutions for all "solve" variables. ≫

$$\begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ p[7][1] & 1 & 0 \\ p[7][1]\, p[11][1] & p[11][1] & 0 \end{pmatrix}$$

```
P = P2 + OreDot[ric, R]
```

$\{\{1, 0, 0, 0\}, \{0, 1, 0, 0\}, \{p[7][1]\, D_t + p[7][1]\, a, -p[7][1]\, (S_t^{-1})\, a\, k + D_t, 0, 0\},$
$\{p[7][1]\, p[11][1]\, D_t + p[7][1]\, p[11][1]\, a,$
$-p[7][1]\, p[11][1]\, (S_t^{-1})\, a\, k + p[11][1]\, D_t, -p[11][1], 1\}\}$

From this idempotent P we can obtain the following decomposition of the system

```
MatrixForm[Decomposition[R, P, A]]
```

$$\begin{pmatrix} -1 & 0 & 0 & 0 \\ 0 & 0 & -(S_t^{-1})\, a\, k & D_t + a \\ 0 & \omega^2 & D_t^2 - p[11][1]\, D_t\, \omega^2 + 2\, D_t\, \omega\, \zeta + \omega^2 & 0 \end{pmatrix}$$